

Assignment 5

Part 1

We choose as example problem the Lotka-Volterra equations - the differential equations describing predator-prey relations as follows:

1. A prey population x increases at a rate $x'[t] = Ax$ (proportional to the number of prey) but is simultaneously destroyed by predators at a rate $x'[t] = -Bxy$ (proportional to the product of the numbers of prey and predators).
2. A predator population y decreases at a rate $y'[t] = -Cy$ (proportional to the number of predators), but increases at a rate $y'[t] = Dxy$ (again proportional to the product of the numbers of prey and predators).

This gives the set of coupled equations:


$$\begin{aligned}x'[t] &= Ax - Bxy \\ y'[t] &= -Cy + Dxy\end{aligned}$$

[<http://mathworld.wolfram.com/Lotka-VolterraEquations.html>]

```

In[113]:= Manipulate[Plot[Evaluate[{x[t], y[t]} /. NDSolve[{x'[t] == a x[t] - b x[t] y[t],
    y'[t] == -c y[t] + d x[t] y[t], x[0] == x0, y[0] == y0}, {x[t], y[t]}, {t, 0, 50}]],
    {t, 0, 50}, PlotRange -> All, AxesLabel -> {"t", "Population"}, ImageSize -> Large],
    {a, 1, 4}, {b, 1, 4}, {c, 1, 4}, {d, 1, 4}, {x0, 5, 20}, {y0, 5, 20}]

```



Part 2

```

In[127]:= SerCos[x_, n_] := Normal[Series[Cos[x1], {x1, 0, n}]] /. x1 -> x
SerSin[x_, n_] := Normal[Series[Sin[x2], {x2, 0, n}]] /. x2 -> x

(* Difference between SerSin and Sin for n=3 *)

```

```

In[141]:= N[SerSin[2, 3] - Sin[2]]

```

```

Out[141]= -0.242631

```

```
In[133]:= N[SerSin[1, 3] - Sin[1]]
```

```
Out[133]:= -0.00813765
```

```
In[142]:= N[SerSin[0.5, 3] - Sin[0.5]]
```

```
Out[142]:= -0.000258872
```

```
In[148]:= N[SerSin[0.01, 3] - Sin[0.01]]
```

```
Out[148]:=  $-8.33332 \times 10^{-13}$ 
```

(*** Difference between SerSin and Sin for n=5 ***)

```
In[144]:= N[SerSin[2, 5] - Sin[2]]
```

```
Out[144]:= 0.0240359
```

```
In[145]:= N[SerSin[1, 5] - Sin[1]]
```

```
Out[145]:= 0.000195682
```

```
In[146]:= N[SerSin[0.5, 5] - Sin[0.5]]
```

```
Out[146]:=  $1.54473 \times 10^{-6}$ 
```


```
In[147]:= N[SerSin[0.01, 5] - Sin[0.01]]
```

```
Out[147]:=  $1.73472 \times 10^{-18}$ 
```

We now plot the $\text{SerSin}[x, 3] - \text{Sin}[x]$ against x .

```
In[150]:= Plot[SerSin[x, 3] - Sin[x], {x, 0, 4}, PlotRange -> All,
  ImageSize -> Large, AxesLabel -> {"x", "SerSin[x,3]-Sin[x"]}]
```

SerSin[x,3]-Sin[x]



```
Out[150]=
```

We now investigate the behavior of $(\text{SerSin}[x, n]^2 + \text{SerCos}[x, n]^2)$. First, consider algebraically:

```
In[163]:= SerSinSq[x_, n_] := Normal[Series[Sin[x1]^2, {x1, 0, n}]] /. x1 -> x
SerCosSq[x_, n_] := Normal[Series[Cos[x2]^2, {x2, 0, n}]] /. x2 -> x
```

```
In[171]:= SerSin[x, 5]
```

```
Out[171]= x -  $\frac{x^3}{6}$  +  $\frac{x^5}{120}$ 
```

```
In[169]:= (* This is the expansion of SerSin[x,3]^2. *)
```

```
In[172]:= Normal[Series[SerSin[x, 5]^2, {x, 0, 10}]]
```

```
Out[172]=  $x^2 - \frac{x^4}{3} + \frac{2x^6}{45} - \frac{x^8}{360} + \frac{x^{10}}{14400}$ 
```

```
In[170]:= (* This is the power series of Sin[x]^2. *)
```


```
In[173]:= SerSinSq[x, 10]
```

```
Out[173]=  $x^2 - \frac{x^4}{3} + \frac{2x^6}{45} - \frac{x^8}{315} + \frac{2x^{10}}{14175}$ 
```

Clearly the series for $\text{Sin}[x]^2$ does not match up with $\text{SerSin}[x, n]^2$. We can observe that this is because squaring the $\text{SerSin}[x, n]$ takes care of all terms correctly up to the n^{th} term - but it also introduces higher order terms that are bound to be inaccurate, since $\text{SerSin}[x, n]$ is missing some terms whose product would sum to such higher order terms in $\text{SerSinSq}[x, n]$, as we can see in the examples above.

We can see the behavior that the error due to the incorrect higher order terms induces in $(\text{SerSin}[x, n]^2 + \text{SerCos}[x, n]^2)$:

```
In[176]:= Plot[SerSin[x, 5]^2 + SerCos[x, 5]^2, {x, 0, 4}, PlotRange -> All,
ImageSize -> Large, AxesLabel -> {"x", "(SerSin[x,5]^2+SerCos[x,5]^2)"}]
```




This confirms our suspicions: as x gets large, the incorrect higher order error terms generate a polynomi-

ally increasing error in $(\text{SerSin}[x, n]^2 + \text{SerCos}[x, n]^2)$.

We now observe and compare the behavior of $(\text{SerSinSq}[x, n] + \text{SerCosSq}[x, n])$, where $\text{SerSinSq}[x, n]$ and $\text{SerCosSq}[x, n]$ are the correct power series for $\sin^2[x]$ and $\cos^2[x]$ respectively:

```
In[184]:= Plot[SerSinSq[x, 5] + SerCosSq[x, 5], {x, 0, 14000}, PlotRange -> All,
  ImageSize -> Large, AxesLabel -> {"x", "(SerSinSq[x,5]+SerCosSq[x,5])"}]
```



As we can see, this sum remains close to the expected value of 1 for a much longer time than the previous series.

Part 3

```
In[185]:= Rx[θ_] := {{1, 0, 0}, {0, Cos[θ], -Sin[θ]}, {0, Sin[θ], Cos[θ]}}
Ry[ξ_] := {{Cos[ξ], 0, Sin[ξ]}, {0, 1, 0}, {-Sin[ξ], 0, Cos[ξ]}}
Rz[φ_] := {{Cos[φ], -Sin[φ], 0}, {Sin[φ], Cos[φ], 0}, {0, 0, 1}}
```

```
In[188]:= Rot3[ψ_, θ_, φ_] := Rz[ψ].Rx[θ].Rz[φ]
```

```
In[198]:= MatrixForm[Rot3[ψ, θ, φ] // Simplify, TableSpacing -> {3, 3}]
```

```
Out[198]/MatrixForm=
```

$$\begin{pmatrix} \cos[\phi] \cos[\psi] - \cos[\theta] \sin[\phi] \sin[\psi] & -\cos[\psi] \sin[\phi] - \cos[\theta] \cos[\phi] \sin[\psi] & \sin[\theta] \sin[\phi] \\ \cos[\theta] \cos[\psi] \sin[\phi] + \cos[\phi] \sin[\psi] & \cos[\theta] \cos[\phi] \cos[\psi] - \sin[\phi] \sin[\psi] & -\cos[\psi] \sin[\theta] \\ \sin[\theta] \sin[\phi] & \cos[\phi] \sin[\theta] & \cos[\theta] \end{pmatrix}$$

```
In[203]:= Rot3Inverse[ψ_, θ_, φ_] := Rz[-φ].Rx[-θ].Rz[-ψ]
```

```
In[204]:= MatrixForm[Rot3Inverse[ψ, θ, φ] // Simplify, TableSpacing → {3, 3}]
```

```
Out[204]//MatrixForm=
```

$$\begin{pmatrix} \cos[\phi] \cos[\psi] - \cos[\theta] \sin[\phi] \sin[\psi] & \cos[\theta] \cos[\psi] \sin[\phi] + \cos[\phi] \sin[\psi] & \sin[\theta] \sin[\psi] \\ -\cos[\psi] \sin[\phi] - \cos[\theta] \cos[\phi] \sin[\psi] & \cos[\theta] \cos[\phi] \cos[\psi] - \sin[\phi] \sin[\psi] & \cos[\phi] \sin[\psi] \\ \sin[\theta] \sin[\psi] & -\cos[\psi] \sin[\theta] & \cos[\theta] \end{pmatrix}$$

```
In[206]:= Rot3[ψ, θ, φ].Rot3Inverse[ψ, θ, φ] // Simplify // MatrixForm
```

```
Out[206]//MatrixForm=
```

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

The product of Rot3 and Rot3Inverse gives us the identity transformation, as expected. We now try an alternative method for finding an inverse, and show that the inverse obtained is equivalent to the one we have found.

```
In[209]:= Inverse[Rot3[ψ, θ, φ]].Rot3[ψ, θ, φ] // Simplify // MatrixForm
```

```
Out[209]//MatrixForm=
```

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

```
In[211]:= Inverse[Rot3[ψ, θ, φ]] - Rot3Inverse[ψ, θ, φ] // Simplify // MatrixForm
```

```
Out[211]//MatrixForm=
```

$$\begin{pmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{pmatrix}$$